# Celestial Navigation

**with**

**Sextant and Calculator HP-42S / Free42**



Ackermans Harry 2023

### Abbreviations and symbols

$LOP1$ = Line of position 1

$LOP2$ = Line of position 2

$COP1$ = Circle of position 1

$COP2$ = Circle of position 2

$H1$ = Observed altitude ($Ho$) in $COP1$

$H2$ = Observed altitude ($Ho$) in $COP2$

$Dec$ = Declination

$GHA$ = Greenwich Hour Angle

$D1 = Dec$ of the sun in $COP1$

$G1 = GHA$ of the sun in $COP1$

$D2 = Dec$ of the sun in $COP2$

$G2 = GHA$ of the sun in $COP2$

$\varphi$ = Latitude of the $FIX$ in the calculations of the COP

$L$ = Longitude of the $FIX$ in the calculations of the COP

$\Delta\varphi$ = Displacement in latitude of the observer

$\Delta L$ = Displacement in longitude of the observer

$lat_M$ = Mean latitude

$latx$ = Dead-reckoning or assumed observer's latitude

$lonx$ = Dead-reckoning or assumed observer's longitude

$FIX$ = Calculated position

$laf = FIX$ latitude

$lof = FIX$ longitude

$IN1, IN2$ = Intercept of LOP1, LOP2

$AZ1, AZ2$ = Azimuth of LOP1, LOP2

$Ha$ = Apparent altitude

$Dmg$ = Distance made Good of the observer

$Cmg$ = Course made Good of the observer

$Ie$ = Index error of the sextant

$Corr$ = Correction of the sextant height according to its certificate

$HoE$ = Height of eye of the observer

$Da.Mo$ = Day and Month of measurement

$Year$ = Year of measurement

$U.T$ = Universal Time of measurement

$ALTs$ = Measured sextant height

$Limb$ = Top edge or bottom edge of the sun

$Npos$ = Northern intersection of the circles of equal altitude

$Spos$ = Southern intersection of the circles of equal altitude

$S.D$ = Semi diameter of the sun

$Tran$ = Noon latitude of the observer (Transit)

$Ho$ = Observed altitude

$GP$ = Geographic Position of the sun

## Foreword

Astronomical positioning by the simultaneous measurement of a number of stars and planets leads to a fairly accurate position, but the practice needed to identify those stars and determine their height above the horizon at twilight will often be lacking. That is why we limit ourselves to determine our location with the help of the sun.
Using nautical almanacs such as the Macmillan and the HO249 tables, we can traditionally determine our position on the basis of sun's altitudes.
With the advent of the programmable calculators, the entire search, calculation and plot work that follows could be taken over from us. The *GHA* and *Dec* of the sun are calculated with a built-in algorithm, as well as the azimuth and intercept of the lines of position. The somewhat difficult chart work is also digitally reproduced, with the final result our position in latitude and longitude *(FIX).*

## HP-42S / Free42

The HP-42S RPN Scientific is a programmable RPN calculator launched by HP in 1988. Its production was discontinued in 1995 but it is still regarded today as one of the best ever made in terms of quality and ease of programming. Its popularity was so great that he is offered on the site of Thomas Okken under the name "**Free42**: An HP-42S Calculator Simulator" as freeware for different operating systems of *computers and smartphones*. The Free42 computer application is delivered in decimal and binary version but it is recommended to install Free42Decimal.exe.
Android and iOS applications too run under the decimal version.
The Free42- app has the advantage that the programs can be saved and loaded as files and shared with other computers. Instructions can be found here. Moreover, the calculation speed is many times higher than with the hardware version of the HP-42S.
Free42 programs are saved with the file extension ".raw".
The programs described below only use a limited number of keys, so that no thorough knowledge of the calculator is required. Those who want to deepen further can go here.

## Programs

This text describes two programs for Free42 and HP-42S to navigate by the sun by means of a sextant or theodolite.
They can be downloaded from the site https://thomasokken.com/free42/42progs/.

*ASTRO_TSO.raw*: calculates the position *(FIX)* using the azimuth-intercept method of Marcq St. Hilaire. This method, developed in 1875, gradually became the standard procedure for astronomical positioning and is based on the choice of an assumed position (A.P), as close as possible to the true position or the estimated position, and the calculation of the intersection of two so-called *lines of position* (LOP further in the text). This program makes the use of plotting sheets superfluous.

*ASTRA_TSO.raw:* calculates the position *(FIX)* using an iteration method that directly calculates the intersections of two *circles of position* (COP further in the text). For a *stationary* observer, contrary to the previous method, no assumed position (A.P) is required so that one can determine his position without any idea of where one is. In the case of a *moving observer (running FIX)*, as in the previous program, the latitude of the observer $latx$ must be fairly well known, especially at higher latitudes, for the calculation of the displacement in longitude $\Delta L$.
Both programs have a built-in almanac of the sun with a maximum deviation of ± 0.5' and an average deviation of ± 0.3' for the *GHA* and *Dec*. For verification, the almanac data on the sites backbearing.com and siranah.de can be consulted.

### Corrections to the measured sextant height $ALTs$

A number of corrections are applied to the measured sextant height in both programs;

- *Instrument correction:* if the sextant is provided with a certificate, this correction can be taken into account for with $Corr$ (in seconds). If not, the input is = 0
- *Index correction:* is a correction for the index error $Ie$ of the sextant which in principle has to be measured before each new observation. $Ie$ is negative if "off the arc" and positive "on the arc".
- *Dip:* this is a correction which is function of the height of eye $HoE$. If an external artificial horizon is used, then $HoE = 0$.

$$Dip['] \approx 1.76 \cdot \sqrt{HoE[m]}$$

- *Refraction:* this is a correction for the atmospheric refraction of the light according to Bennett's formula.

$$Refraction['] = \frac{1}{\tan\left(Ha[°] + \frac{7.31}{Ha[°] + 4.4}\right)}$$

- *Parallax:* for the parallax in altitude a correction based on the horizontal parallax of the sun (0.144') is applied.
- *Semi diameter (S.D):* this correction is made if the bottom or top edge *(Limb)* of the sun is measured. The *S.D* is calculated based on the mean anomaly *(AM)* of the sun with the formula

$$S.D['] = \frac{16.02}{1 - 0.017 \,.\cos AM}$$

**Sextant heights**

$Ha = ALTs + Corr - Ie - Dip$ (Apparent altitude)

$Ho = Ha - Refraction + Parallax \pm S.D$ (Observed altitude)

$HoE = 0$ implies the use of an external artificial horizon

If an external artificial horizon is used, then $Dip = 0$ and $Ha = (ALTs + Corr - Ie)/2$

**Used formulas**

- If the $Cmg$ and $Dmg$ are known then the observer's displacement in latitude and longitude are respectively:

$$\Delta\varphi[°] = Dmg[nm] \,.\, \cos Cmg\,/60$$

$$\Delta L[°] = \frac{Dmg[nm] \,.\, \sin Cmg}{\cos lat_M \,\,.\, 60}$$

in which the mean latitude

$$lat_M = latx + \Delta\varphi/2$$

and $latx$ is the (assumed) latitude before displacement.

- In *ASTRO_TSO.raw* (azimuth-intercept method) the *FIX*-coordinates are

$$laf = \frac{IN2 \,.\, \sin AZ1 - IN1 \,.\, \sin AZ2}{\sin(AZ1 - AZ2)} + latx$$

$$lof = \frac{IN1 \,.\, \cos AZ2 - IN2 \,.\, \cos AZ1}{\sin(AZ1 - AZ2) \,.\, \cos latx} + lonx$$

where $latx$ and $lonx$ are the coordinates of the A.P after displacement. These values are calculated from the A.P before displacement, $Cmg$ and $Dmg$.

**Format and units of the input data**

| | | |
|---|---|---|
| *latx, lonx* | | + (N and E) – (S and W) |
| *Ie* | x.xx [minutes] | – OFF the arc + ON the arc |
| *HoE* | xx.xx [meter] | Ext. artificial horizon HoE=0 |
| *Da.Mo* | xx.xx [day.month] | Ex: 5.07 23.12 |
| *U.T* | xx.xx xx [h.min sec] | Ex: 17.0533 6.0806 |
| *latx, lonx, ALTs* | xxx.xx x [ddd°mm.m’] | Ex: –15.416 56.093 |
| *Limb* | Lower=–1 Center=0 Upper=1 | |
| *Dmg* | xxx.x [nautical miles] | |
| *Cmg* | xxx.x [°] | |
| *Corr* | xx [seconds] | Ex: –17 12 |

**Program components**

***ASTRO_TSO.raw***: consists of 3 subroutines; *LOP1, LOP2*, and *TRAN*

- *LOP1 (Line of position1):* here the assumed position (A.P), which is usually the dead-reckoning position, and the data of the first observation are entered. The subroutine calculates the intercept *IN1* and azimuth *AZ1* of the first line of position *LOP1* and stores them in their variables with the same name.

- *LOP2 (Line of position2):* on the basis of the *Dmg* and *Cmg* a new A.P is defined. Together with the data of the second observation, the *IN2* and *AZ2* of *LOP2* are calculated relative to this new A.P. The coordinates of the intersection of the “offset” *LOP1* and *LOP2*, being the *FIX*, are obtained with the above formulas. From a geometric point of view this actually amounts to the translation of the *LOP1* in the direction (=*Cmg*) of the new A.P over a distance (=*Dmg*). The coordinates of the *FIX* are stored as $laf$ and $lof$.
  Since the intercept method ignores the curvatures of the actual position lines, the obtained *FIX* is not our exact position but rather an improved position (compared to AP). The residual error remains tolerable as long as the radii of the circles of equal altitude are not too small and the AP is not too far from our actual position. The geometric error inherent to the intercept method can be decreased by iteration, i.e., substituting the obtained *FIX* for AP and repeating the calculations with the same altitudes and GP’s (see appendix). This will result in a more accurate position. If necessary, we can reiterate the procedure until the obtained position remains virtually constant (rarely needed).

- *TRAN (Transit):* this part calculates the noon latitude of our position.
  The input of the estimated longitude $lonx$ only results in a rough estimate of the time of transit (LAN = Local Apparent Noon) as it is calculated on the basis of a fixed transit time of the Greenwich meridian (12h U.T) without taking into account the EoT (Equation of Time).
  It is more accurate to start from the actual "Greenwich meridian transit time" and to add 4 minutes per degree of western longitude or subtract 4 minutes per degree of eastern longitude. The input data refer to the moment of culmination of the sun. "South" and "North" indicate the position of the sun relative to the observer. The noon latitude is stored as $laf$.

Notes

- *LOP1* and *LOP2* must be entered chronologically.

- No Dip correction may be applied when using a **bubble sextant** and/or a **theodolite**. To make the distinction with the use of an **external** artificial horizon *(HoE = 0)* and to prevent *Ha* from being divided by 2, set ***HoE = 0.0001*** and the Dip correction is negligible.

- Because of their simplicity, the **mean latitude** formulas are often used in everyday navigation. Mean latitude is a good approximation for rhumb line navigation for short and medium distances. The method is less suitable for polar regions (convergence of meridians).

***ASTRA_TSO.raw*** : consists of 3 subroutines; *COP1, COP2,* and *TRAN*

- *COP1 (Circle of position1):* $latx$ is the estimated or known latitude of the first observation and will be used in *COP2* to calculate the displacement in longitude $\Delta L$. For a stationary observation (*Dmg=0*) this value is irrelevant.
  After inputting the other measured values, this routine calculates the parameters that define the *Circle of position1*, i.e. *GHA (= G1), Dec (= D1)* and *Ho (= H1).*

- *COP2 (Circle of position2):* with $Dmg$ and $Cmg$ we calculate the displacement of the first *circle of position COP1*, being $\Delta\varphi$ in latitude and $\Delta L$ in longitude. With the other measured values, this routine calculates the parameters that define the *Circle of position2*, i.e.
  *GHA (= G2), Dec (= D2), Ho (= H2)*, and ultimately our position *(FIX).*
  These 2 circles normally have 2 intersections on the globe. These are our 2 possible positions. When calculating the coordinates of these points of intersection, the choice is made between the northern intersection point ($Npos$) and the southern intersection point ($Spos$). The coordinates of the *FIX* are stored as $laf$ and $lof$.
  The algorithm for calculating the intersections is discussed in detail in the appendix.

- *TRAN (Transit):* see *ASTRO_TSO.raw*

Notes: see *ASTRO_TSO.raw*

***ASTRO* versus *ASTRA***

- If one is completely in the dark about his position one can not enter a suitable A.P into ASTRO. In case of a *stationary measurement (Dmg=0)*, ASTRA does not need these data. With 2 observations, preferably with a difference in azimuth between 30° and 150°, an accurate position can be calculated apart from the normal measurement errors.




- The iteration process of the ASTRA-program runs very fast in Free42 but can take a lot of time (several seconds) with the HP-42S due to its low clock speed.

- ASTRA always iterates to the most accurate position. With ASTRO, at a great distance between the A.P and the true position and / or at large sextant heights, at least 1 iteration may be required with the calculated FIX as the new A.P.

**Free42 and HP-42S commands**

- Press the XEQ button. The list of programs that have been loaded appears in the display.

- Press the key below the name of the selected program and after viewing the label, press R/S. The main menu appears in the display, consisting of the 3 subroutines and the EXIT routine to close the program.




- Each routine is started with the key under its name. The requested data are entered with the numeric keys. Pay particular attention to the units and the format. For negative numbers, first enter the number followed by +/-. An incorrect input can be deleted with the ← key and with RTN you return to the main menu. Each input is followed by R/S. In a selection menu, press the key below the desired choice. If the program stops with relevant information in the display, press R/S to continue.

- Entered data and calculated values are stored as variables and can be called up with RCL and scroll keys ▲ and ▼

- *GHA, Dec, laf, lof, latx* and *lonx* are stored in the format [ddd°mm.m']. *AZ1*, *IN1* and *S.D* in [°].

- The results of the observations can be printed out. The print function is activated by PRINT ▲ PON and disabled by PRINT ▲ POFF. The print is available for almost all operating systems via the File menu. On the iPhone the screen must be touched just below the status bar or by swiping to the left.

**Appendix: calculation of the intersections of 2 circles of position**

We apply the formulas of the nautical triangle to 2 observations;

$$\sin H1 = \sin D1 \,.\sin(\varphi - \Delta\varphi) + \cos D1 \,.\cos(\varphi - \Delta\varphi)\,.\cos(G1 + L - \Delta L) \qquad (1)$$

$$\sin H2 = \sin D2 \,.\sin\varphi + \cos D2 \,.\cos\varphi \,.\cos(G2 + L) \qquad (2)$$

(1) and (2) are the equations of 2 circles of position in which $\varphi$ and $L$ are the geographical coordinates of the position of the observer at the time of the second measurement, i.e. the *FIX*.
$H1$ and $H2$ are the true sextant heights (after corrections) and $\Delta\varphi$ and $\Delta L$ are respectively the displacement in latitude and longitude after the first measurement.

The system of 2 equations with 2 unknowns ($\varphi$ and $L$) cannot be solved analytically, but still by iteration. In order for this process to converge to a result for $\varphi$ and $L$
*- the iteration must be started in the equation of the circle with the smallest sextant height $H$, i.e. the circle with the largest radius.*
*- since for 1 value of $\varphi$, 2 values of $L$ are possible and vice versa, the progam makes the following choice:*

If *H2 ≤ H1*

then the starting value of $\varphi$ (= D2) is placed in equation (2). This gives 2 possible $L$-values:
$L = Acos(\ldots) - G2$ and $L = -Acos(\ldots) - G2$. The program chooses the most eastern value and since Acos is always positive

$$L = Acos\left(\frac{\sin H2 - \sin D2\,.\,\sin\varphi}{\cos D2\,.\cos\varphi}\right) - G2$$

This L-value is placed in (1) and the new $\varphi$-value for (2) is calculated

$$\varphi = 2.Atan\left(\frac{\sin D1 \pm \sqrt{\sin^2 D1 - \sin^2 H1 + \cos^2 D1\ .\cos^2(G1 + L - \Delta L)}}{\sin H1 + \cos D1\,.\cos(G1 + L - \Delta L)}\right) + \Delta\varphi$$

with $+\sqrt{\dots}$ for the northern intersection and $-\sqrt{\dots}$ for the southern intersection.

If *H2 > H1*

then the starting value of $\varphi$ (= D2) is placed in equation (1). This gives 2 possible $L$ -values: $L = Acos(\dots) - G1 + \Delta L$ and $L = -Acos(\dots) - G1 + \Delta L$. The program chooses the most western value and since Acos is always positive

$$L = -Acos\left(\frac{\sin H1 - \sin D1\,.\,\sin(\varphi - \Delta\varphi)}{\cos D1\,.\cos(\varphi - \Delta\varphi)}\right) - (G1 - \Delta L)$$

This L-value is placed in (2) and the new $\varphi$-value for (1) is calculated

$$\varphi = 2.Atan\left(\frac{\sin D2 \pm \sqrt{\sin^2 D2 - \sin^2 H2 + \cos^2 D2\,.\cos^2(G2 + L)}}{\sin H2 + \cos D2\,.\cos(G2 + L)}\right)$$

with $+\sqrt{\dots}$ for the northern intersection and $-\sqrt{\dots}$ for the southern intersection.

In both cases the value of φ is calculated by substitution, by means of the so-called T-formulas, with variables $\tan((\varphi - \Delta\varphi)/2)$ and $\tan(\varphi/2)$ respectively.

The iterations are stopped when 2 consecutive values of $\varphi$ differ by less than 0.001°

The choices made in the iteration process described above implicitly assume that the GP of the sun at the second observation is to the west of the GP of the first observation. If it is to the east, the program will adjust these choices.

**Appendix: iterations in ASTRO**

For a stationary measurement *(Dmg=0)*, the calculated position can be refined by inputting the *FIX* as the new A.P with the following key sequence in the main menu:

LOP1 RCL LAF R/S RCL LOF R/S and repeating the calculations with the initial input data.

In case of a "running FIX" the new A.P is calculated with the following key sequence:

LOP1 RCL LAF R/S RCL LOF R/S RTN LOP2 R/S 180 + R/S RTN

Starting at LOP1 the calculations are repeated with the initial input data without changing the new calculated A.P values $latx$ and $lonx$.

**List of equations used in the text**

$$\sin H1 = \sin D1 \,.\, \sin(\varphi - \Delta\varphi) + \cos D1 \,.\, \cos(\varphi - \Delta\varphi) \,.\, \cos(G1 + L - \Delta L)$$

$$\sin H2 = \sin D2 \,.\, \sin\varphi + \cos D2 \,.\, \cos\varphi \,.\, \cos(G2 + L)$$

$$L = Acos\left(\frac{\sin H2 - \sin D2 \,.\, \sin\varphi}{\cos D2 \,.\, \cos\varphi}\right) - G2$$

$$\varphi = 2.Atan\left(\frac{\sin D1 \pm \sqrt{\sin^2 D1 - \sin^2 H1 + \cos^2 D1 \,.\, \cos^2(G1 + L - \Delta L)}}{\sin H1 + \cos D1 \,.\, \cos(G1 + L - \Delta L)}\right) + \Delta\varphi$$

$$L = -Acos\left(\frac{\sin H1 - \sin D1 \,.\, \sin(\varphi - \Delta\varphi)}{\cos D1 \,.\, \cos(\varphi - \Delta\varphi)}\right) - (G1 - \Delta L)$$

$$\varphi = 2.Atan\left(\frac{\sin D2 \pm \sqrt{\sin^2 D2 - \sin^2 H2 + \cos^2 D2 \,.\, \cos^2(G2 + L)}}{\sin H2 + \cos D2 \,.\, \cos(G2 + L)}\right)$$

$$\Delta\varphi[°] = Dmg[nm] \,.\, \frac{\cos Cmg}{60}$$

$$\Delta L[°] = \frac{Dmg[nm] \,.\, \sin Cmg}{\cos lat_M \,.\, 60}$$

$$lat_M = latx + \Delta\varphi/2$$

$$laf = \frac{IN2 \,.\, \sin AZ1 - IN1 \,.\, \sin AZ2}{\sin(AZ1 - AZ2)} + latx$$

$$lof = \frac{IN1 \,.\, \cos AZ2 - IN2 \,.\, \cos AZ1}{\sin(AZ1 - AZ2) \,.\, \cos latx} + lonx$$

$$Dip['] \approx 1.76 \cdot \sqrt{HoE[m]}$$

$$Refraction['] = \frac{1}{\tan\left(Ha[°] + \frac{7.31}{Ha[°] + 4.4}\right)}$$

$$Parallax['] = 0.144 \cdot \cos(Ha - Refraction)$$

$$S.D['] = \frac{16.02}{1 - 0.017 \,.\, \cos AM}$$

**Interesting websites**

http://thomasokken.com/free42/
https://www.celnav.de/page2.htm
http://www.teacupnavigation.net/CN.html
http://www.backbearing.com/almanac.htm
https://www.siranah.de/html/sail040p.htm

**Miscellaneous**

- A running fix is not as accurate as a stationary fix. Course (*Cmg*) and distance (*Dmg*) over ground can only be estimated since current and wind (drift) are not exactly known in most cases.

- Moreover, there is an optimum range of altitudes (*ALTs*) the navigator should choose to obtain reliable results. Low altitudes increase the influence of abnormal refractions (random error), whereas high altitudes, corresponding to circles of equal altitude with small diameters (and large curvatures), increase the geometric error due to their equalization with a line of position (*LOP*). The latter does not apply to the *ASTRA* program. The recommended range is 20°- 70°.

- Also the horizon dip (*Dip*) can be affected by atmospheric refraction and become unpredictable under certain meteorological conditions. This does not apply when using an artificial horizon as no correction for *Dip* on the sextant height (*ALTs*) is applied in this case.

- To limit the random errors, two series of measurements of the sun are preferably made with a difference in azimuth between 30 and 150° (see "A short guide to Celestial Navigation", page 16-1, on the site of H. Umland, https://www.celnav.de/page2.htm) and the arithmetic average of the times (*U.T*) and the corresponding sextant heights (*ALTs*) are used to calculate a *LOP* or *COP*. (see the “Average Shots” program attached)

- To measure the index error (*Ie*) in the absence of a natural horizon, the procedure described on the site https://www.nauticalalmanac.it/en/navigation-astronomy can be followed.

- Observing with an artificial horizon: instead of making the reflected sun via the sextant coincide with the directly reflected sun, the "upper and lower limb" is used for greater accuracy (see "Celestial Navigation in a Teacup" , page 50, on the site http://www.teacupnavigation.net/CN.html).

- After calculating a “running fix”, the data of *LOP2* and *COP2* respectively, together with the coordinates of the *FIX*, can be copied to *Line of position1 (LOP1)* and *Circle of position1 (COP1)* respectively with the following key sequence:

For *ASTRO*: LOP1 RCL LAF R/S RCL LOF R/S …. plus the remaining input data (Ie, date, U.T, Alts, etc) are the data from the first observation for the next running fix.

For *ASTRA*: COP1 RCL LAF R/S …. plus the remaining input data (Ie, date, U.T, Alts, etc), are the data from the first observation for the next running fix.

Entering the data from a second observation in *LOP2* and *COP2* respectively, results in a new running fix.

- *ASTRO* and *ASTRA* provide a default printout if the print function is activated. Additional information can be added to the print at any time via PRINT, after the format of the display has been defined via DISP. Usually FIX 03 suffices. With R/S you return to the program.

- Below the program steps of *ASTRO_TSO.raw*, *ASTRA_TSO.raw* and *AVSHO.raw*

# ASTRONAVIGATION - SUN HP 42S

```
001 LBL "ASTRO"
002 DEG
003 CF 02
004 CF 03
005 CF 29
006 "......□.ASTRO.□"
007 ├"If.........zon"
008 AVIEW
009 STOP

010 LBL 00
011 "LOP1"
012 KEY 1 XEQ A
013 "LOP2"
014 KEY 2 XEQ B
015 "TRAN"
016 KEY 4 XEQ C
017 "EXIT"
018 KEY 6 GTO J
019 CLA
020 AVIEW
021 MENU
022 STOP
023 GTO 00

024 LBL A
025 XEQ 01
026 FC? 01
027 RTN
028 STO IN1   (°)
029 RCL 04
030 STO AZ1   (°)
031 XEQ D
032 RTN

033 LBL B
034 XEQ 10
035 FIX 01
036 INPUT Dmg
037 INPUT Cmg
038 COS
039 RCLx Dmg
040 60
041 ÷
042 STO 23          =Δφ
043 RCL 18
044 +
045 XEQ 09
046 STO lat*
047 RCL Cmg
048 SIN
049 RCLx Dmg
050 RCL 18
051 RCL 23
052 2
053 ÷
054 +
055 COS
056 ÷
057 60
058 ÷
059 RCL 17
060 +
061 XEQ 09
062 STO lon*
063 XEQ 02
064 FC? 01
065 RTN
066 XEQ D
067 XEQ 03
068 SF 02
069 XEQ D
070 PRA
071 PROMPT
072 RTN

073 LBL C
074 XEQ 10
075 FIX 03
076 INPUT lon*
077 XEQ 08
078 15
079 ÷
080 12
081 X<>Y
082 -
083 >HMS
084 "Transit.."
085 PRA
086 ├".~."
087 AIP
088 ├":"
089 FP
090 100
091 x
092 AIP
093 ├".U.T"
094 PROMPT
095 XEQ 02
096 FC? 01
097 RTN

098 LBL c
099 RCL 01
100 90
101 RCL- 08
102 "SOUTH"
103 KEY 1 GTO 14
104 "NORTH"
105 KEY 4 GTO 13
106 CLA
107 MENU
108 PROMPT
109 GTO c

110 LBL 13
111 ±

112 LBL 14
113 +
114 XEQ 09
115 STO laf
116 SF 02
117 SF 03
118 XEQ D
119 CF 03
120 PRA
121 PROMPT
122 RTN

123 LBL 01
124 XEQ 10
125 FIX 03
126 INPUT lat*
127 XEQ 08
128 STO 18
129 INPUT lon*
130 XEQ 08
131 STO 17

132 LBL 02
133 FIX 02
134 INPUT Ie
135 INPUT HoE
136 XEQ I
137 INPUT Da.Mo
138 IP
139 STO 00          =JO
140 LAST X
141 FP
142 100
143 x
144 STO 15          =MO
145 RCL IND ST x
146 STO+ 00
147 FIX 00
148 INPUT Year
149 4
150 ÷
151 FP
152 STO 02          =AA
153 X=0?
154 XEQ a
155 FIX 04
156 INPUT U.T
157 >HR
158 24
159 ÷
160 STO+ 00        =J2
161 FIX 03
162 INPUT ALTs
163 FIX 00
164 INPUT Corr
165 INPUT Limb
166 RCL Year
167 2000
168 -
169 365.25
170 x
171 0.5
172 +
173 RCL- 02
174 365250
175 ÷
176 STO 14
177 RCL 00
178 LAST X
179 ÷
180 +
181 STO 01          =T
182 RAD
183 6283.01961
184 x
185 0.043179665
186 -
187 STO 02          =AM
188 XEQ 20
189 0.033417
190 RCL 02
191 SIN
192 x
193 RCL+ 02
194 RCL 02
```

| | | |
|---|---|---|
| 195 | 2 | |
| 196 | x | |
| 197 | SIN | |
| 198 | 0.0003489 | |
| 199 | x | |
| 200 | + | |
| 201 | RCL 01 | |
| 202 | 0.300052641 | |
| 203 | x | |
| 204 | 4.938242632 | |
| 205 | + | |
| 206 | + | |
| 207 | STO 03 | |
| 208 | SIN | |
| 209 | 0.397777 | |
| 210 | x | |
| 211 | ASIN | |
| 212 | >DEG | |
| 213 | STO 01 | =DEC |
| 214 | XEQ 09 | |
| 215 | STO DEC | |
| 216 | RCL 03 | |
| 217 | TAN | |
| 218 | 0.917482 | |
| 219 | x | |
| 220 | ATAN | |
| 221 | STO 02 | |
| 222 | RCL 03 | |
| 223 | COS | |
| 224 | X<0? | |
| 225 | XEQ b | |
| 226 | RCL 00 | |
| 227 | 360.98564735 | |
| 228 | x | |
| 229 | RCL 14 | |
| 230 | 360007.7 | |
| 231 | x | |
| 232 | 100.465 | |
| 233 | + | |
| 234 | 360 | |
| 235 | MOD | =TSO |
| 236 | + | |
| 237 | 360 | |
| 238 | ÷ | |
| 239 | FP | |
| 240 | PI | |
| 241 | x | |
| 242 | 2 | |
| 243 | x | |
| 244 | RCL- 02 | |
| 245 | >DEG | =GHA |
| 246 | X<0? | |
| 247 | XEQ 12 | |
| 248 | ENTER | |
| 249 | XEQ 09 | |
| 250 | STO GHA | |
| 251 | R↓ | |
| 252 | DEG | |
| 253 | RCL lon* | |
| 254 | XEQ 08 | |
| 255 | STO 05 | |
| 256 | + | |
| 257 | STO 02 | =LHA |
| 258 | RCL lat* | |
| 259 | XEQ 08 | |
| 260 | STO 06 | |
| 261 | COS | |
| 262 | RCL 01 | |
| 263 | SIN | |
| 264 | x | |
| 265 | RCL 06 | |
| 266 | SIN | |
| 267 | RCL 01 | |
| 268 | COS | |
| 269 | x | |
| 270 | RCL 02 | |
| 271 | COS | |
| 272 | x | |
| 273 | - | |
| 274 | RCL 01 | |
| 275 | COS | |
| 276 | ± | |
| 277 | RCL 02 | |
| 278 | SIN | |
| 279 | x | |
| 280 | X<>Y | |
| 281 | >POL | McMillan 159 |
| 282 | X<>Y | |
| 283 | X<0? | |
| 284 | XEQ 12 | |
| 285 | STO 04 | =AZ |
| 286 | RCL 01 | |
| 287 | SIN | |
| 288 | RCL 06 | |
| 289 | SIN | |
| 290 | x | |
| 291 | RCL 01 | |
| 292 | COS | |
| 293 | RCL 02 | |
| 294 | COS | |
| 295 | x | |
| 296 | RCL 06 | |
| 297 | COS | |
| 298 | x | |
| 299 | + | |
| 300 | ASIN | |
| 301 | STO 02 | =Hc |
| 302 | RCL ALTs | |
| 303 | XEQ 08 | |
| 304 | RCL Ie | |
| 305 | 60 | |
| 306 | ÷ | |
| 307 | - | |
| 308 | RCL Corr | |
| 309 | 3600 | |
| 310 | ÷ | |
| 311 | + | |
| 312 | RCL HoE | |
| 313 | X=0? | Artif. Hor |
| 314 | GTO 16 | |
| 315 | SQRT | |
| 316 | 0.0293 | =Dip |
| 317 | x | |
| 318 | - | |
| 319 | LBL 15 | |
| 320 | STO 03 | |
| 321 | 4.4 | =Refrac |
| 322 | + | |
| 323 | 7.31 | |
| 324 | X<>Y | |
| 325 | ÷ | |
| 326 | RCL+ 03 | |
| 327 | TAN | |
| 328 | 1/X | |
| 329 | 60 | |
| 330 | ÷ | |
| 331 | RCL 03 | |
| 332 | X<>Y | |
| 333 | - | |
| 334 | ENTER | =Parallax |
| 335 | COS | |
| 336 | 0.0024 | |
| 337 | x | |
| 338 | + | |
| 339 | RCL Limb | |
| 340 | RCL S.D | |
| 341 | x | |
| 342 | - | |
| 343 | STO 08 | =Ho |
| 344 | RCL- 02 | |
| 345 | STO 03 | =INT |
| 346 | RTN | |
| 347 | LBL 16 | |
| 348 | R↓ | |
| 349 | 2 | |
| 350 | ÷ | |
| 351 | GTO 15 | |
| 352 | LBL a | |
| 353 | 1 | |
| 354 | STO 02 | |
| 355 | RCL 15 | |
| 356 | 2 | |
| 357 | X>=Y? | |
| 358 | RTN | |
| 359 | 1 | |
| 360 | STO+ 00 | |
| 361 | RTN | |
| 362 | LBL b | |
| 363 | PI | |
| 364 | STO+ 02 | |
| 365 | RTN | |
| 366 | LBL 03 | =FIX |
| 367 | "If.......*.FIX*" | |
| 368 | AVIEW | |
| 369 | RCL AZ1 | |
| 370 | SIN | |
| 371 | RCLx 03 | |
| 372 | RCL 04 | |
| 373 | SIN | |
| 374 | RCLx IN1 | |
| 375 | - | |
| 376 | RCL AZ1 | |
| 377 | RCL- 04 | |
| 378 | SIN | |
| 379 | STO 07 | |
| 380 | ÷ | |
| 381 | RCL+ 06 | |
| 382 | XEQ 09 | |
| 383 | STO laf | |
| 384 | RCL 04 | |
| 385 | COS | |
| 386 | RCLx IN1 | |
| 387 | RCL AZ1 | |
| 388 | COS | |
| 389 | RCLx 03 | |
| 390 | - | |
| 391 | RCL: 07 | |
| 392 | RCL 06 | |
| 393 | COS | |
| 394 | ÷ | |
| 395 | RCL+ 05 | |
| 396 | ABS | |
| 397 | 180 | |
| 398 | X<Y? | |
| 399 | GTO 11 | |
| 400 | LAST X | |
| 401 | LBL 04 | |
| 402 | XEQ 09 | |
| 403 | STO lof | |
| 404 | RTN | |
| 405 | LBL D | |

```
CLMENU
CLA
RCL Da.Mo
AIP
├"/"
FP
100
x
AIP
├"/"
FIX 00
ARCL Year
├".□."
RCL U.T
AIP
├":"
FP
100
x
AIP
├":"
FP
100
x
AIP
FS? 02
GTO d
├"IfAZ:."
ARCL 04
├"°.INT:."
RCL 03
60
x
FIX 01
ARCL STx
├""
PRA
PROMPT
RTN

LBL d

CF 02
├"If"
RCL Iaf
ENTER
XEQ 07
R↓
X<0?
GTO 05
├"N.."

LBL e

FS? 03
RTN
RCL Iof
ENTER
XEQ 07
R↓
X<0?
GTO 06
├"E"
RTN

LBL 05

├"S.."
GTO e

LBL 06

├"W"
RTN

LBL 07

ABS
AIP
├"°"
FP
100
x
FIX 01
RND
ARCL ST x
RTN

LBL 08     =>d.ddd

IP
LAST X
FP
5
x
3
÷
+
RTN

LBL 09=>dd°mm.m

IP
LAST X
FP
3
x
5
÷
+
RTN

LBL 10

CLMENU
SF 01
"RTN"
KEY 6 GTO H
MENU
RTN

LBL 11

LAST X
ENTER
SIGN
360
x
-
GTO 04

LBL 12

360
+
RTN

LBL H

CF 01
RTN

LBL I
-1
STO 01
30
ST0 02
58
STO 03
89
STO 04
119
STO 05
150
STO 06
180
STO 07
211
STO 08
242
STO 09
272
STO 10
303
STO 11
333
STO 12
RTN

LBL 20      =>S.D

COS
0.017
x
1
X<>Y
-
0.267
X<>Y
÷
STO S.D
RTN

LBL J

CLMENU
EXITALL
CLST
DEG
FIX 04
END
```

Sight Reduction with 2 Lines of Position (with sextant or theodolite) and calculator HP 42S or App FREE42

| | |
|---|---|
| lat*, lon* = + (N en E) | HP 42S 1369 bytes |
| - (S en W) | App Free42 1363 bytes |
| HoE = meter | Artificial Horizon HoE=0 |
| Limb | lower= -1 centre= 0 upper= 1 |
| Format lat*, lon*, ALTs | = xxx.xx x [ ddd°mm.m' ] |
| Format U.T | = xx.xx xx [ h . min sec ] |
| Format Da.Mo | = xx.xx [ Day . Month ] |
| Dmg | = xxx.x [ nautical miles ] |
| Cmg | = xxx.x [ ° ] |
| Index error Ie | = - OFF the arc + ON the arc |
| Format Ie | = x.xx [ minutes ] |
| Format Corr | = xx [ sec ] |

Vers. 12/04/2020






```
LBL "ASTRA"

DEG
CF 02
CF 03
CF 05
CF 29
"......□.ASTRA.□"
├ "If.........zon"
AVIEW
STOP

LBL 00

"COP1"
KEY 1 XEQ A
"COP2"
KEY 2 XEQ B
"TRAN"
KEY 4 XEQ C
"EXIT"
KEY 6 GTO J
CLA
AVIEW
MENU
STOP
GTO 00

LBL A

XEQ 01
FC? 01
RTN
STO 20        =SIN H1
RCL 00
STO 21        =GHA1
RCL 01
STO 22        =DEC1
XEQ D
RTN

LBL B

XEQ 10
FIX 01
INPUT Dmg
INPUT Cmg
COS
RCLx Dmg
60
÷
STO 23        =Δφ
RCL Cmg
SIN
RCLx Dmg
RCL 19
RCL 23
2
÷
+
COS        =Mean Lat
÷
60
÷
STO 24        =ΔL
XEQ 12
FC? 01
RTN
STO 02        =SIN H2
XEQ 02
FC? 01
RTN
RCL 11
XEQ 09
STO laf
RCL 10
ABS
180
X<Y?
GTO 11
LAST X

LBL b

XEQ 09
STO lof
SF 02
XEQ D
PRA
PROMPT
RTN

LBL C        =Transit

XEQ 10
FIX 03
INPUT lon*
XEQ 08
15
÷
12
X<>Y
-
>HMS
"Transit.."
PRA
├ ".~."
AIP
├ ":"
FP
100
x
AIP
├ ".U.T"
PROMPT
XEQ 12
FC? 01
RTN

LBL c

RCL 01
90
RCL- 08
"SOUTH"
KEY 1 GTO 16
"NORTH"
KEY 4 GTO 15
CLA
MENU
PROMPT
GTO c

LBL 15

±

LBL 16

+
XEQ 09
STO laf
SF 02
SF 03
XEQ D
CF 03
PRA
PROMPT
RTN

LBL 01

XEQ 10
FIX 03
INPUT lat*        =DRP
XEQ 08
STO 19

LBL 12

FIX 02
INPUT Ie
INPUT HoE
XEQ I
INPUT Da.Mo
IP
STO 00        =JO
LAST X
FP
100
x
STO 15        =MO
RCL IND ST x
STO+ 00
FIX 00
INPUT Year
4
÷
FP
STO 02        =AA
X=0?
XEQ 13
FIX 04
INPUT U.T
>HR
24
÷
STO+ 00        =J2
FIX 03
INPUT ALTs
RCL Year
2000
-
365.25
x
0.5
+
RCL- 02
365250
÷
STO 14
RCL 00
LAST X
÷
+
STO 01        =T
RAD
6283.01961
x
0.043179665
-
STO 02        =AM
XEQ 20
```

```
193 0.033417
194 RCL 02
195 SIN
196 x
197 RCL+ 02
198 RCL 02
199 2
200 x
201 SIN
202 0.0003489
203 x
204 +
205 RCL 01
206 0.300052641
207 x
208 4.938242632
209 +
210 +
211 STO 03
212 SIN
213 0.397777
214 x
215 ASIN
216 >DEG
217 STO 01       =DEC
218 XEQ 09
219 STO DEC
220 RCL 03
221 TAN
222 0.917482
223 x
224 ATAN
225 STO 02
226 RCL 03
227 COS
228 X<0?
229 XEQ 14
230 RCL 00
231 360.98564735
232 x
233 RCL 14
234 360007.7
235 x
236 100.465
237 +
238 360
239 MOD          =TSO
240 +
241 360
242 ÷
243 FP
244 PI
245 x
246 2
247 x
248 RCL- 02
249 >DEG
250 X<0?
251 XEQ 24
252 STO 00       =GHA
253 XEQ 09
254 STO GHA
255 DEG
256 FIX 00
257 INPUT Corr
258 INPUT Limb
259 RCL ALTs
260 XEQ 08
261 RCL Ie
262 60           =Ie
263 ÷
264 -
265 RCL Corr
266 3600
267 ÷
268 +
269 RCL HoE
270 X=0?         Artif.Hor
271 GTO 18
272 SQRT
273 0.0293
274 x            =Dip
275 -

276 LBL 17

277 STO 03
278 4.4          =Refract
279 +
280 7.31
281 X<>Y
282 ÷
283 RCL+ 03
284 TAN
285 1/X
286 60
287 ÷
288 RCL 03
289 X<>Y
290 -
291 ENTER        =Parallax
292 COS
293 0.0024
294 x
295 +
296 RCL Limb
297 RCL S.D
298 x
299 -
300 STO 08       =Ho
301 SIN
302 RTN
303 LBL 13

304 1
305 STO 02
306 RCL 15
307 2
308 X>=Y?
309 RTN
310 1
311 STO+ 00
312 RTN

313 LBL 14

314 PI
315 STO+ 02
316 RTN

317 LBL 18

318 R↓
319 2
320 ÷
321 GTO 17

322 LBL 02

323 CF 04
324 "Npos"
325 KEY 1 GTO 22
326 "Spos"
327 KEY 4 GTO 21
328 CLA
329 MENU
330 PROMPT
331 GTO 02

332 LBL 21

333 SF 04

334 LBL 22

335 CF 05
336 RCL 00
337 RCL- 21
338 ABS
339 180
340 X>Y?
341 GTO 19
342 RCL 00
343 RCL 21
344 X<Y?
345 SF 05
346 LBL 23

347 ".......*.FIX*"
348 AVIEW
349 RCL 01
350 STO 05       =STARTφ
351 SIN
352 STO 06
353 LAST X
354 COS
355 STO 07       =COS D2
356 RCL 21       =G1
357 RCL- 24
358 STO 04       =G1- ΔL
359 RCL 22
360 SIN
361 STO 08       =SIN D1
362 LAST X
363 COS
364 STO 09       =COS D1
365 RCL 02
366 RCL 20
367 X<Y?
368 GTO 04

369 LBL 03

370 RCL 02       =SIN H2
371 RCL 06       =SIN D2
372 RCL 05
373 SIN
374 x
375 -
376 RCL÷ 07
377 RCL 05
378 COS
379 ÷
380 ACOS
381 FS? 05
382 ±
383 RCL- 00
384 CPX?
385 GTO 04
386 STO 10       =L
387 RCL+ 04
388 COS
389 RCLx 09
390 STO 12
391 X↑2
392 RCL 20
393 X↑2
394 -
395 RCL 08
396 X↑2
397 +
398 X<0?
```

```
GTO 04
SQRT
FS? 04
±
RCL+ 08
RCL 20
RCL+ 12
÷
ATAN
2
x
RCL+ 23
STO 11          =φ
RCL- 05
ABS
0.001
X>=Y?
RTN
RCL11
STO 05
GTO 03

LBL 04

RCL 20     =SIN H1
RCL 08     =SIN D1
RCL 05
RCL- 23
STO 12
SIN
x
-
RCL 12
COS
÷
RCL÷ 09
ACOS
FC? 05
±
RCL- 04
CPX?
GTO 03
STO 10          =L
RCL+ 00     =G2+L
COS
RCLx 07
STO 12
X↑2
RCL 02
X↑2
-
RCL 06
X↑2
+
X<0?
GTO 03
SQRT
FS? 04
±
RCL+ 06
RCL 02
RCL+ 12
÷
ATAN
2
x
STO 11          =φ
RCL- 05
ABS
0.001
X>=Y?
RTN
RCL11
STO 05
GTO 04

LBL D

CLA
RCL Da.Mo
AIP
├ "/"
FP
100
x
AIP
├ "/"
FIX 00
ARCL Year
├ ".□."
RCL U.T
AIP
├ ":"
FP
100
x
AIP
├ ":"
FP
100
x
AIP
FS? 02
GTO d
├ "IfALTs:."
FIX 03
ARCL ALTs
PRA
PROMPT
RTN
LBL d

CF 02
├ "If"
RCL Iaf
ENTER
XEQ 07
R↓
X<0?
GTO 05
├ "N.."

LBL e

FS? 03
RTN
RCL Iof
ENTER
XEQ 07
R↓
X<0?
GTO 06
├ "E"
RTN

LBL 05

├ "S.."
GTO e

LBL 06

├ "W"
RTN

LBL 07

ABS
AIP
├ "°"
FP
100
x
FIX 01
RND
ARCL ST x
RTN

LBL 08     =>d.ddd

IP
LAST X
FP
5
x
3
÷
+
RTN

LBL 09 =>dd°mm.m

IP
LAST X
FP
3
x
5
÷
+
RTN

LBL 10

CLMENU
SF 01
"RTN"
KEY 6 GTO H
MENU
RTN

LBL 11

LAST X
ENTER
SIGN
360
x
-
GTO b

LBL H

CF 01
RTN

LBL I

-1
STO 01
30
ST0 02
58
STO 03
89
STO 04
119
STO 05
150
STO 06
180
STO 07
```

```
596  211
597  STO 08
598  242
599  STO 09
600  272
601  STO 10
602  303
603  STO 11
604  333
605  STO 12
606  RTN

607  LBL 19

608  RCL 00
609  RCL 21
610  X>Y?
611  SF 05
612  GTO 23

613  LBL 20        =>S.D

614  COS
615  0.017
616  x
617  1
618  X<>Y
619  -
620  0.267
621  X<>Y
622  ÷
623  STO S.D
624  RTN

625  LBL 24

626  360
627  +
628  RTN

629  LBL J

630  CLMENU
631  EXITALL
632  CLST
633  DEG
634  FIX 04
635  END
```

Sight Reduction with 2 Circles of Position (with sextant or theodolite) and calculator HP 42S or App FREE42

| | |
|---|---|
| lat*, lon* = + (N en E) | HP 42S 1452 bytes |
| - (S en W) | App Free42 1446 bytes |
| HoE = meter | Artificial Horizon HoE=0 |
| Limb | lower= -1 centre= 0 upper= 1 |
| Format lat*, lon*, ALTs | = xxx.xx x [ ddd°mm.m' ] |
| Format U.T | = xx.xx xx [ h . min sec ] |
| Format Da.Mo | = xx.xx [ Day . Month ] |
| Dmg | = xxx.x [ nautical miles] |
| Cmg | = xxx.x [ ° ] |
| Index error Ie | = - OFF the arc + ON the arc |
| Format Ie | = x.xx [ minutes ] |
| Format Corr | = xx [ sec ] |

Vers. 12/04/2020



## AVERAGE SHOTS HP 42S / Free42

```
LBL "AVSHO"

DEG
CF29
"....Average"
├".ShotsIf"
PROMPT
"Avera"
KEY 1 GTO A
"Reset"
KEY 4 GTO 01
"EXIT"
KEY 6 GTO J
MENU

LBL 01

0
STO "Time"
STO "Hs"
STO 01
STO 02
STO 03

LBL 02

FIX 04
INPUT "Time"
>HR
STO+ 01
FIX 03
INPUT "Hs"
XEQ 08
STO+ 02
1
STO+ 03
GTO 02

LBL A

RCL 01
RCL 03
÷
>HMS
STO "AvTim"
RCL 02
RCL 03
÷
XEQ 09
STO "AvHs"
VIEW "AvTim"
STOP
VIEW "AvHs"
STOP
GTO A

LBL 08

IP
LASTX
FP
5
x
3
÷
+
RTN

LBL 09

IP
LASTX
FP
3
x
5
÷
+
RTN

LBL J

CLMENU
EXITALL
CLST
CLV "Hs"
CLV "Time"
FIX 04
END
```

Calculation of the arithmetic mean of the times (AvTim) and the sextant heights (AvHs) of a set of observations (shots) within a short period of time (a few minutes) to limit "random errors".

| | |
|---|---|
| Format "Time" en "AvTim" = | xx.xx xx [ h . min sec ] |
| Format "Hs" en "AvHs" = | xxx.xx x [ ddd°mm.m' ] |
| App Free42 196 bytes | HP 42S 196 bytes |

Vers. 26/05/2020